

Агрегатные функции СКД, о которых мало кто знает

17.02.24

Разработка - СКД


Пользуетесь ли Вы всеми возможными агрегатными функциями, которые предоставляет система компоновки данных? Если Вы используете только: СУММА, КОЛИЧЕСТВО, МИНИМУМ, МАКСИМУМ, СРЕДНЕЕ, то эта статья для Вас.

Предыдущие мои статьи по СКД можно посмотреть по ссылкам ниже:

- Основные понятия и элементы схемы компоновки
- Настройки вариантов отчетов
- Работа с пользовательскими настройками
- Расширения языка запросов
- Наборы данных и связи между ними, иерархии
- Внутренние функции СКД, ВычислитьВыражение

Нет такого программиста 1С, который никогда не работал бы с СКД. И все, кто выводил с помощью системы компоновки даже самые простые отчеты, знают, что такое агрегатные функции. Так как вывод итогов по ресурсам на уровне группировок, это, пожалуй, самая распространенная задача при создании отчета.

Но, далеко не все знают, что список возможных агрегатных функций не ограничивается выпадающим списком в конструкторе СКД:



Если посмотреть справку то можно увидеть гораздо больше агрегатных функций, о существовании некоторых из которых мало кто знает:

Агрегатные функции

[Сумма](#)
[Количество](#)
[Количество различных](#)
[Максимум](#)
[Минимум](#)
[Среднее](#)
[Массив](#)
[ТаблицаЗначений](#)
[Свернуть](#)
[ПолучитьЧасть](#)
[Упорядочить](#)
[СоединитьСтроки](#)
[ГрупповаяОбработка](#)
[Каждый](#)
[Любой](#)
[СтандартноеОтклонениеГенеральнойСовокупности](#)
[СтандартноеОтклонениеВыборки](#)
[ДисперсияВыборки](#)
[ДисперсияГенеральнойСовокупности](#)
[КовариацияГенеральнойСовокупности](#)
[КовариацияВыборки](#)
[Корреляция](#)
[РегрессияНаклон](#)
[РегрессияОтрезок](#)
[РегрессияКоличество](#)
[РегрессияR2](#)
[РегрессияСреднееX](#)
[РегрессияСреднееY](#)
[РегрессияSXX](#)
[РегрессияSYX](#)
[РегрессияSYY](#)
[МестоВПорядке](#)
[КлассификацияABC](#)

А в выпадающем списке консоли СКД отображены только основные агрегатные функции, которые чаще всего приходится применять.

Но мы, программисты 1С довольно редко читаем справку. Обычно мы пытаемся понять новый материал интуитивно, максимум обращаемся к синтакс-помощнику, и очень зря. Ведь, использование этих функций позволяет сильно облегчить и, главное, ускорить создание большого количество отчетов.

Статья как раз про эти, «скрытые» агрегатные функции, которые иногда оказываются ну очень полезными. В статье будут описаны возможности этих функций, по некоторым будут приведены примеры.

Массив

Функция формирует массив, содержащий значения детальных записей по полю или выражению, указанному в качестве параметра. Обработываются подчиненные записи текущей группировки. Ключевое слово «Различные», позволяет получить только уникальные записи. Если в качестве параметра передана таблица значений, сформируется массив со значениями первой колонки таблицы:

Массив(Различные Номенклатура)

Массив(Количество)

Если массив формируется по числовому полю, то к его результату можно применять еще одну агрегатную функцию (Сумма, Количество, Максимум, Минимум, Среднее).

ТаблицаЗначений

Функция формирует таблицу значений. Количество колонок результирующей функции равно количеству переданных параметров (полей или выражений). Обрабатываются подчиненные записи текущей группировки. После ключевого слова «КАК» можно задать наименования колонок таблицы. Ключевое слово «Различные» перед первым параметром позволяет получить только различные записи. Проверка выполняется по всем колонкам таблицы значений.

ТаблицаЗначений(Различные Контрагент КАК Клиент, Номенклатура КАК Товар)

Если в качестве параметров указаны поля-остатки и данные выбираются с периодичностью (Остатки и обороты), строки таблицы формируются в разрезе периодов и уникальной комбинации измерений. То-есть, если значение остатка в разных периодах различное, в итоговой таблице значений будет сформировано несколько строк, со значениями из разных периодов.

Для наглядности приведу пример.

Создадим набор данных со следующим запросом:

```
ВЫБРАТЬ
    ТоварныеЗапасыОстаткиИОбороты.КоличествоНачальныйОстаток,
    ТоварныеЗапасыОстаткиИОбороты.КоличествоКонечныйОстаток,
    ТоварныеЗапасыОстаткиИОбороты.Товар КАК Товар,
    ТоварныеЗапасыОстаткиИОбороты.Склад КАК Склад
ИЗ
    РегистрНакопления.ТоварныеЗапасы.ОстаткиИОбороты(, , месяц, , ) КАК ТоварныеЗапасыОстаткиИОбороты
```

Добавляем вычисляемое поле «ТаблицаЗначений» и в качестве выражения ресурса, укажем для него:

ТаблицаЗначений(ПериодМесяц КАК Период, Склад КАК Склад, КоличествоКонечныйОстаток КАК Остаток)

В результате получаем таблицу значений со значениями на каждый день, в котором было изменение остатка:

Товар Склад	Количество Конечный остаток	Таблица значений
Молоко	608,00	02.11.2017 0:00:00; Склад отдела продаж; 100 04.10.2017 0:00:00; Малый; 200 02.11.2017 0:00:00; Малый; 400 14.11.2017 0:00:00; Малый; 500 25.02.2018 0:00:00; Большой; 20 27.02.2018 0:00:00; Большой; 8
Большой	8,00	25.02.2018 0:00:00; Большой; 20 27.02.2018 0:00:00; Большой; 8
Малый	500,00	04.10.2017 0:00:00; Малый; 200 02.11.2017 0:00:00; Малый; 400 14.11.2017 0:00:00; Малый; 500
Склад отдела продаж	100,00	02.11.2017 0:00:00; Склад отдела продаж; 100
Итого	608,00	02.11.2017 0:00:00; Склад отдела продаж; 100 04.10.2017 0:00:00; Малый; 200

Важное дополнение. Если для колонок таблицы значений не заданы наименования после ключевого слова «КАК», во всех функциях, которые выполняют обработку этой таблицы (Свернуть, Упорядочить, ПолучитьЧасть) нельзя использовать наименования для указания требуемых колонок. Имена колонок будут иметь специальные наименования, назначенные системой. Для обращения к этим колонкам

можно будет использовать только их порядковые номера. Так что если планируется обращение к колонкам таблицы значений по именам, обязательно нужно указывать эти имена после ключевого слова «КАК».

Свернуть

Функция позволяет удалить повторяющиеся строки из массива или таблицы значений. Если в качестве первого параметра используется Таблица значений, необходимо вторым параметром указать порядковые номера или имена колонок, по которым нужно свернуть данные:

```
Свернуть(Массив(Товар))
Свернуть(ТаблицаЗначений(Номенклатура КАК Товар, КоличествоОборот) , "Товар")
Свернуть(ТаблицаЗначений(КоличествоОборот) , "1")
```

Важный момент, данная функция не сворачивает данные с группировкой суммовых показателей. Происходит именно отсечение строк с повторяющимися значениями:

	Покупатель	Количество	Свернуть
	Товар	о Оборот	
+	Магазин "Бытовая техника"	6,00	5
			1
+	Магазин "Мясная лавка"	1 912,00	1 265
			260
			375
			12
-	Магазин "Обувь"	6,00	3
	Ботинки	3,00	3
	Валенки	3,00	3
+	Магазин "Продукты"	1 585,00	318
			112
			505
			650

Для массива, функция аналогична применению ключевого слова «Различные» в функции «Массив». Для таблицы значений, функция позволяет проверить уникальность **не по всем колонкам**, в отличие от применения ключевого слова «Различные» в функции «ТаблицаЗначений».

ПолучитьЧасть

Функция позволяет получить часть таблицы значений. Оставить только необходимые колонки. Наименования или порядковые номера колонок нужно указать вторым параметром функции:

```
ПолучитьЧасть(ТаблицаЗначений(Номенклатура, Количество) , "1")
ПолучитьЧасть(ТаблицаЗначений(Номенклатура КАК Товар, Количество) , "Товар")
```

В результате получится таблица значений с одной колонкой «Номенклатура».

Упорядочить

Функция позволяет упорядочить массив или таблицу значений. Если используется таблица значений, вторым параметром нужно указать номера колонок, по которым требуется выполнить упорядочивание. Можно указать направление упорядочивания (Возр / Убыв) и признак «Автоупорядочивание»:

```
Упорядочить (Массив(Номенклатура))
Упорядочить(ТаблицаЗначений(Номенклатура, Количество), "1 Возр Автоупорядочивание")
Упорядочить(ТаблицаЗначений(Номенклатура КАК Товар, Количество), "Товар Убыв")
```

СоединитьСтроки

Функция выполняет соединение строк, переданных в качестве массива или таблицы значений. Втором параметром можно указать символ соединения строк, массива или таблицы (по умолчанию используется символ перевода строки). Если в качестве первого параметра используется таблица значений, в третьем параметре можно указать символ соединения колонок таблицы (по умолчанию используется символ «;»). Результатом выполнения функции является строка.

Примеры использования функций:

```
СоединитьСтроки(Массив(Товар), ";")
СоединитьСтроки(ТаблицаЗначений(Товар, КоличествоОборот), ";", ",", ",")
```

	Покупатель Товар	Количество Оборот	Соединить строки
[-]	Магазин "Бытовая техника"	6,00	VekoNT02,5;Veko345MO,1
	Veko345MO	1,00	Veko345MO,1
	VekoNT02	5,00	VekoNT02,5
[-]	Магазин "Мясная лавка"	1 912,00	Хлеб,1 265;Торт ,260;Колбаса,375;Молоко,12
	Колбаса	375,00	Колбаса,375
	Молоко	12,00	Молоко,12
	Торт	260,00	Торт ,260
	Хлеб	1 265,00	Хлеб,1 265
[-]	Магазин "Обувь"	6,00	Валенки,3;Ботинки,3
	Ботинки	3,00	Ботинки,3
	Валенки	3,00	Валенки,3
[-]	Магазин "Продукты"	1 585,00	Хлеб,318;Колбаса,112;Торт ,505;Йогурт,650
	Йогурт	650,00	Йогурт,650
	Колбаса	112,00	Колбаса,112
	Торт	505,00	Торт ,505
	Хлеб	318,00	Хлеб,318

ГрупповаяОбработка

Функция выполняет групповую обработку строк. Вычисляет указанные выражения и формирует таблицу значений с результатами этих выражений в контексте текущей или указанной группировки. Важный момент, что в обработку включаются строки, которые находятся на одном уровне с обрабатываемой строкой или на одном уровне иерархии (это не подчиненные строки как для большинства агрегатных функций).

Синтаксис функции следующий:

```
ГрупповаяОбработка(Выражения, ВыраженияИерархии, ИмяГруппировки)
```

Выражения – поля или выражения, которые требуется обработать и поместить в результирующие данные.

ВыраженияИерархии – поля или выражения, которые требуется обработать для иерархических записей (если обработка выполняется для иерархии). Если параметр не указан, используются такие же поля, которые указаны в параметре «*Выражения*».

ИмяГруппировки – Имя группировки, по строкам которой необходимо выполнить расчет. Если параметр не указан, вычисление выполняется по строкам на уровне текущей группировки. Если указанная группировка не доступна на текущем уровне (группировка более низкого уровня), функция вернет значение NULL.

Результат групповой обработки возвращается в виде объекта с типом «*ДанныеГрупповойОбработкиКомпоновкиДанных*». Данные групповой обработки можно передать в качестве параметра в экспортную функцию общего модуля

Объект содержит следующие реквизиты:

- Данные – результирующая таблица значений. Колонками таблицы являются поля или результаты выражений, перечисленные в первом параметре функции «*ГрупповаяОбработка*».
- ТекущийЭлемент – строка таблицы значений соответствующая текущему элементу группировки.
- ВременныеДанныеОбработки – структура в которую можно поместить произвольные параметры в процессе работы с результатом групповой обработки в функции общего модуля. Эти параметры будут доступны при следующей обработке групповых данных в рамках этой же группировки.

Пример использования функции

Создадим набор данных с простым запросом:

ВЫБРАТЬ

ПродажиОбороты.Покупатель КАК Покупатель,
ПродажиОбороты.Товар КАК Товар,
ПродажиОбороты.КоличествоОборот КАК КоличествоОборот,
ПродажиОбороты.СуммаОборот КАК СуммаОборот

ИЗ

РегистрНакопления.Продажи.Обороты КАК ПродажиОбороты

Создадим вычисляемое поле «*ГрупповаяОбработка*», определим его как ресурс и укажем следующее выражение:

СКДСервер.ФункцияОбщегоМодуля(ГрупповаяОбработка ("СУММА(КоличествоОборот) КАК Количество"))

В конфигурации, в общем серверном модуле «*СКДСервер*» разместим экспортную процедуру:

Функция ФункцияОбщегоМодуля(Данные) Экспорт

```

ВременныеДанные = Данные.ВременныеДанныеОбработки;
Если ВременныеДанные.Свойство("СреднееКоличество") Тогда
    СреднееКоличество = ВременныеДанные.СреднееКоличество;
Иначе
    ОбщееКоличество = 0;
    Для каждого СтрокаДанных Из Данные.Данные Цикл
        ОбщееКоличество = ОбщееКоличество + СтрокаДанных.Количество;
    КонецЦикла;
    СреднееКоличество = Округ(ОбщееКоличество / Данные.Данные.Количество());
    ВременныеДанные.Вставить("СреднееКоличество", СреднееКоличество);
КонецЕсли;

Если Данные.ТекущийЭлемент <> Неопределено Тогда
    ОтклонениеОтСреднего = СреднееКоличество - Данные.ТекущийЭлемент.Количество;
Иначе
    ОтклонениеОтСреднего = 0;
КонецЕсли;

Возврат ОтклонениеОтСреднего;

```

КонецФункции

Функция рассчитывает отклонение значения в текущей группировке от среднего значения по всем группировкам.

Проверяется наличие параметра «СреднееКоличество» в структуре «**ВременныеДанныеОбработки**». Если параметр существует, используется он, так как среднее количество в рамках одной и той же группировки будет одинаковым. Если параметра не существует (выводится первая строка группировки), выполняется его расчет по таблицы значений «**Данные**». Полученное значение добавляется в структуру «**ВременныеДанныеОбработки**».

На заметку. Имейте в виду про следующую особенность. Функция общего модуля, указанная для ресурса, вызывается дважды для одной и той же строки. Это связано с типовой обработкой расшифровки для поля – ресурса. Если для данного поля создать собственный макет, с указанием одного и того же выражения для основного параметра и для параметра расшифровки, такой особенности не будет. Функция будет вызвана один раз для каждой строки.

Каждый

Функция вычисляет выражения для всех строк и возвращает значение «Истина», если выражения по всем строкам вернули значение «Истина». В противном случае, будет возвращено значение «Ложь».

Каждый(КоличествоОборот > 10)

	Покупатель Товар	Количество о Оборот	Каждый
[-]	Магазин "Бытовая техника"	6,00	Нет
	Veiko345MO	1,00	Нет
	VeikoNT02	5,00	Нет
[-]	Магазин "Мясная лавка"	1 912,00	Да
	Колбаса	375,00	Да
	Молоко	12,00	Да
	Торт	260,00	Да
	Хлеб	1 265,00	Да
[-]	Магазин "Обувь"	6,00	Нет
	Ботинки	3,00	Нет
	Валенки	3,00	Нет
[-]	Магазин "Продукты"	1 585,00	Да
	Йогурт	650,00	Да
	Колбаса	112,00	Да
	Торт	505,00	Да
	Хлеб	318,00	Да

Любой

Функция вычисляет выражения для всех строк и возвращает значение «*Истина*», если хотя бы по одной строке результат будет «*Истина*». В противном случае, будет возвращено значение «*Ложь*».

Любой(КоличествоОборот > 10)

МестоВПорядке

Функция определяет позицию данной строки, если расположить все строки группировки в порядке, указанном в выражении первого параметра. Также как и для функции «*ГрупповаяОбработка*», используются строки, которые находятся на одном уровне с обрабатываемой или на одном уровне иерархии (не вложенные строки).

Строки с одинаковым значением порядка будут иметь одно и то же место.

Синтаксис функции следующий:

МестоВПорядке(Порядок, ПорядокИерархии, ИмяГруппировки)

Порядок – выражение порядка, по которому происходит сортировка строк. Можно указать направление упорядочивания (Возр / Убыв) и признак «*Автоупорядочивание*».

ПорядокИерархия – выражение упорядочивания для иерархических записей. Если не указано, действует выражение для основных записей.

ИмяГруппировки – группировка, в контексте которой необходимо выполнить упорядочивание. Если параметр не указан, порядок вычисляется в контексте текущей группировки.

Примеры использования:

```
МестоВПорядке("СУММА(КоличествоОборот)")
МестоВПорядке("Номенклатура Возр Автоупорядочивание")
МестоВПорядке("СУММА(КоличествоОборот) ", СУММА("СуммаОборот"))
```

Для обычных строк группировки действует сортировка по полю «КоличествоОборот», для иерархических – «СуммаОборот»:

Для обычных записей порядок по полю «КоличествоОборот»

	Товар	Количество о Оборот	Сумма Оборот	Место в поряд ке
[-]	Обувь	15,00	36 600,00	2
	Босоножки	5,00	12 500,00	4
	Ботинки	3,00	6 000,00	1
	Валенки	3,00	4 500,00	1
	Сапоги	4,00	13 600,00	3
[+]	Продукты	3 797,00	424 895,00	4
[+]	Услуги	2,00	600,00	1
[+]	Электротовары	16,00	99 600,00	3
	Итого	3 830,00	561 695,00	

Для иерархических записей порядок по полю «СуммаОборот»

```
МестоВПорядке("Сумма(КоличествоОборот)",, "Покупатель")
```

Для всех вложенных группировок место в порядке будет рассчитано по группировкам верхнего уровня «Покупатель»:

	Покупатель	Количество Оборот	Место в порядке2
	Товар		
[-]	Магазин "Бытовая техника"	6,00	2
	Veko345MO	1,00	2
	VekoNT02	5,00	2
[-]	Магазин "Мясная лавка"	1 912,00	6
	Колбаса	375,00	6
	Молоко	12,00	6
	Торт	260,00	6
	Хлеб	1 265,00	6
[+]	Магазин "Обувь"	6,00	2
[+]	Магазин "Продукты"	1 585,00	5
[+]	Попов Б.В. ИЧП	316,00	4
[-]	Шлюзовая ООО	5,00	1
	Sony K3456P	3,00	1
	Veko67NE	1,00	1
	Доставка	1,00	1
	Итого	3 830,00	

КлассификацияABC

Функция вычисляет значение классификации «ABC» по переданному выражению и пороговым значениям.

Синтаксис функции следующий:

```
КлассификацияABC(Значение, КоличествоГрупп, ПроцентыДляГрупп, ИмяГруппировки)
```

Значение – поле или выражение, для которого требуется рассчитать классификацию ABC.

КоличествоГрупп – количество групп классификации. Можно указать количество больше трех, тогда уже будет классификация «ABCD», «ABCDE» и т.д.

ПроцентыДляГрупп – пороги для попадания в группы через запятую в виде строки. Причем для последней группы порог не указывается. Диапазон последней группы считается от последнего порога до 100%. Пример указания порогов для трех групп: «50, 80».

ИмяГруппировки - группировка, в контексте которой необходимо выполнить расчет. Если параметр не указан, вычисление происходит в контексте текущей группировки.

Примеры использования:

КлассификацияABC("Сумма(КоличествоОборот)", 4, "40, 70, 90")

Товар	Количество о Оборот	Классификация ABC
Bosch1234	2,00	4
Bosch15	3,00	4
Sony K3456P	4,00	4
Veko345MO	1,00	4
Veko67NE	1,00	4
VekoNT02	5,00	4
Босоножки	5,00	4
Ботинки	3,00	4
Валенки	3,00	4
Доставка	2,00	4
Йогурт	950,00	2
Колбаса	487,00	3
Молоко	12,00	4
Сапоги	4,00	4
Торт	765,00	2
Хлеб	1 583,00	1

КлассификацияABC("Сумма(КоличествоОборот)", 3, "50, 80", "Покупатель")

Классификация вложенных строк выполняется по строкам группировки верхнего уровня «Покупатель»:

Покупатель	Количество о Оборот	Классификация ABC
Магазин "Бытовая техника"	6,00	3
Veko345MO	1,00	3
VekoNT02	5,00	3
Магазин "Мясная лавка"	1 912,00	1
Колбаса	375,00	1
Молоко	12,00	1
Торт	260,00	1
Хлеб	1 265,00	1
Магазин "Обувь"	6,00	3
Ботинки	3,00	3
Валенки	3,00	3
Магазин "Продукты"	1 585,00	1
Попов Б.В. ИП	316,00	2
Шлюзовая ООО	5,00	3

В СКД существует еще целый набор математических агрегатных функций:

- СтандартноеОтклонениеГенеральнойСовокупности
- СтандартноеОтклонениеВыборки
- ДисперсияВыборки
- ДисперсияГенеральнойСовокупности

- КовариацияГенеральнойСовокупности
- КовариацияВыборки
- Корреляция
- РегрессияНаклон
- РегрессияОтрезок
- РегрессияКоличество
- РегрессияR2
- РегрессияСреднееX
- РегрессияСреднееY
- РегрессияSXX
- РегрессияSYY
- РегрессияSXY

Их рассмотрение и использование заслуживает отдельной статьи, так что здесь я их разбирать не буду.

Спасибо за внимание. В следующих статьях я планирую **рассматривать программную работу с отчетами на СКД**, а также привести разбор некоторых интересных примеров.

Для ознакомления с предыдущими статьями можно перейти по ссылкам ниже:

- Основные понятия и элементы схемы компоновки
- Настройки компоновки данных
- Пользовательские настройки
- Расширения языка запросов для СКД
- Наборы данных и связи между ними, собственная иерархия
- Внутренние функции СКД, примеры использования



INFOSTART TOOLKIT

Самый продвинутый набор инструментов для управляемых форм 1С 8.3: Консоль запросов, Консоль СКД, Консоль кода, Редактор объекта, Анализ прав доступа, Метаданные, Поиск ссылок, Сравнение объектов, Подписки на события и др.

[УЗНАТЬ ПОДРОБНЕЕ](#)

Вступайте в нашу телеграмм-группу Инфостарт

Инфостарт бот

